

```

1 *****
2 * A Practical Guide to Using Panel Data
3 * Simonetta Longhi and Alita Nandi
4 * ISER, University of Essex
5 * Chapter 9
6 *****
7
8 version 12
9 clear all
10 set more off
11 capture log close
12
13 cd "C:\My Documents"
14
15 global dir "C:\My Documents"
16
17 log using "$dir\Example_Chapter9.log", replace
18
19 * 9.1. Introduction
20 *-----
21
22 use "$dir/DataFile", clear
23
24 * 9.2.1. Pooled Estimations
25
26 generate LnW = ln(paygu)
27 generate age2 = age^2
28 recode sex (1 = 0) (2 = 1), gen(Female)
29
30 label list rmastat
31 tabulate mastat
32 generate Married = 1 if mastat == 1 | mastat == 2 | mastat == 7
33 replace Married = 0 if (mastat >= 3 & mastat <= 6 ) ///
34 | (mastat >= 8 & mastat <= 10)
35
36 tabulate qfachi, gen(Q)
37
38 tabulate region, gen(R)
39 * Note that the BHPS has two variables for regions
40 * We have included in the dataset the variable region (Chapter 4)
41 * while we have used region2 in Chapter 7
42 * For the purpose of this analysis either of the two variable is ok
43
44 tabulate wave, gen(Y)
45
46 regress LnW age age2 Female Married Q1-Q6 R1-R6 R8-R19 Y1-Y17, vce(cluster pid)
47 estimates store R_OLS
48
49 * 9.2.2. Individual Unobserved Heterogeneity
50
51 xtset pid wave, yearly
52
53 xtreg LnW age age2 Female Married Q1-Q6 R1-R6 R8-R19 Y1-Y17, fe
54 estimates store R_FE
55
56 xtreg LnW age age2 Female Married Q1-Q6 R1-R6 R8-R19 Y1-Y17, re
57 estimates store R_REgls
58
59 xtreg LnW age age2 Female Married Q1-Q6 R1-R6 R8-R19 Y1-Y17, mle
60 estimates store R_REmle
61
62 foreach var of varlist LnW age age2 Married Q1-Q6 R1-R6 R8-R19 Y1-Y17 {
63     generate D`var' = `var' - L.`var'
64 }
65 regress DLnW Dage Dage2 DMarried DQ1-DQ6 DR1-DR6 DR8-DR19 DY1-DY17
66 estimates store R_FD
67
68 * 9.2.3. Choosing the Right Estimator
69
70 hausman R_FE R_REgls
71
72
73 * 9.4. Further Estimators
74
75 foreach var of varlist age age2 Married Q1-Q6 R1-R6 R8-R19 Y1-Y17 {

```

```

76     bysort pid: egen M`var' = mean(`var')
77     }
78     xtreg LnW age age2 Female Married Q1-Q6 R1-R6 R8-R19 Y1-Y17    ///
79     Mage Mage2 MMarried MQ1-MQ6 MR1-MR6 MR8-MR19 MY1-MY17, re
80
81
82
83
84
85     * Save the results in a table *
86     * See Chapter 14             *
87     *****
88
89     * Table 9.1
90     quietly estout R_OLS R_FE R_FD R_RE*                        ///
91     using "$dir\Wages.out",                                     ///
92     cells(b(star fmt(%9.3f)) se(par fmt(%9.3f)))               ///
93     style(tab) stats(r2 N, fmt(%9.3f %9.0g))                   ///
94     labels(R2 Observations) label collabels(, none)           ///
95     starlevels(+ 0.05 * 0.01)                                  ///
96     postfoot("St.err. in parenthesis; + Significant 5%, * Significant 1%") ///
97     replace
98
99     log close
100

```